
ABSTRACT 
Creating electronic documents in PDF file format is becoming increasingly popular nowadays. Combining ODS PDF 
statement and the REPORT procedure in SAS can create various PDF output files with different styles. During the 
process of validating those data in PDF file, there is demand to import PDF summary tables or listings into SAS 
datasets. A utility is developed which reads in a SAS generated uncompressed PDF file, extracts and converts the 
data from the PDF file into a SAS datasets. The overview of this utility is presented in this paper. 

INTRODUCTION  
PDF stands for "portable document format".  It was introduced to ease the sharing of printable documents between 
computers across operating system platforms when one needs to save files that cannot be modified. For the detail, 
see reference section <PDF Reference, version 1.7 - Adobe>. In the biotech and many other industries, RTF and  

PDF output files have been used extensively to present reports and analysis.  Many involve importing RTF data into 
SAS datasets but not much has been done for PDF data due to raised level of complexity and difficulty in parsing 
PDF formats. 

By default, statement ODS PDF usually generates a compressed PDF file with default setting ‘compress=6’.  This 
paper talks about importing SAS generated uncompressed PDF (‘compress=0’) output into SAS data set by the utility 
macro %PDF2SAS which is introduced in the following sections.  This macro reads in the whole uncompressed PDF 
file as strings using INFILE and INPUT statements in a DATA step.  Perl Regular Expression function separates and 
extracts targeted information, including text strings and attributes for each cell.  The final step is to identify each cell in 
terms of the row and column locations and concatenate multiple wrapped text strings together to form a complete cell 
when needed.  %PDF2SAS is a light weight macro which imports PDF data to a SAS data set quickly and efficiently. 

OVERVIEW OF THE PROGRAM 
SAMPLE TABLE 

An uncompressed PDF file can be easily generated in statement ODS PDF with option COMPRESS=0 which is 
shown on display 1. In statement ODS PDF, a style option can be set as: STYLE=<style name>. In Appendix1, the 
SAS code demonstrates how to create an uncompressed PDF file. Following is the output PDF file 
“afmsg_Analysis.pdf”: 

 
Display 1: Uncompressed output PDF file which is created by ODS PDF and PROC REPORT. 

http://www.adobe.com/content/dam/Adobe/en/devnet/acrobat/pdfs/pdf_reference_1-7.pdf

To see how different the PDF file’s format is, let’s open it in NOTEPAD which is a basic text editor in Windows OS. 
The following shows the format of cell with “IO_CAN_NOT_OPEN” in the table.  

In PDF file afmsg_Analysis.pdf: 

 “BT /TT2 9.5 Tf 86.88 527.52 Td (IO_CAN_NOT_OPEN)Tj ET” 

We can see that the ‘IO_CAN_NOT_OPEN’ is embedded in some numbers and characters. Those numbers and 
characters represent the attributes of the cell ‘IO_CAN_NOT_OPEN ‘. 

The text ‘IO_CAN_NOT_OPEN is placed at 527.52 points (7.33 inches) from the bottom of the page and 86.88 points 
(1.21 inches) from the left edge, using 9.5 point Arial font. Point is a unit and usually used in typography, computers 
font sizes and printing as the smallest unit. The abbreviation is "pt". 1 Inch = 72 Points. 

A typical format for text in PDF is as follows: 

1. BT: Begin a text object. 

2. /TT2 9.5 Tf: Set the font and font size to use, installing them as parameters in the text state. The font size is 9.5 
point. The font resource identified by the name TT2 specifies the font externally known as Arial, according to the 
following string inside of the file: obj<</Type/Font/Subtype/TrueType/Name/TT2/BaseFont/CCUAAB+Arial,Regular/… 

3. 86.88 527.52 Td: Specify a starting position on the page, setting parameters in the text object. 

4. (IO_CAN_NOT_OPEN)Tj: Paint the glyphs for a string of characters at that position. 

5. ET:  End the text object. 

Therefore, when reading in the whole PDF file the first and most important step is to extract the useful text and 
attributes from the long codes into SAS data. 

 

PROGRAM FLOW AND DETAILS 
Now, we start to introduce the code of the macro. The interface of the macro is very simple. There are 2 input 
parameters – ‘datafile’ and ‘out’. 
 
%macro pdf2sas( 
  datafile = /* Specifies the complete path and filename for the input PDF file     */ 
, out     = /* Identifies the output SAS data set with a one or two-level SAS name */ 

); 

 

The program flow of the pdf2sas macro can roughly be outlined as follows: 

1. Validate the parameter’s values and check if the input parameters are valid. 

2. Read the PDF file into SAS and then extract each relevant text and attribute as a string. 

3. Separate text and attributes from the string. 

4. Obtain row and column number based on the text and their attributes. 

 5. Re-construct text from data set when the text is wrapped into a cell as multiple line 
 

1. VALIDATING THE PARAMETER’S VALUES 

In the beginning of the macro, the code checks the validity of input parameters.  
Namely the existence of the pdf file name and the lib reference of output SAS dataset.  

%local _msg_ _lin_; 
%let datafile = %scan(&datafile,1,.); 
 
%if %sysfunc(fileexist(&datafile..pdf))=0 %then 
 %let _msg_ = In parameter DATAFILE, PDF file "&datafile..pdf" does not 
exist.; 
%else %if %sysfunc(libref(%scan(%scan(&out,-2,.) work,1)))^=0 %then 
%let _msg_ = In parameter OUT, libname "%scan(&out,1,.)" is not assigned.; 
%if %length(&_msg_)>0 %then %do; 

%ER:%let _lin_ = %sysfunc(repeat(=,%length(%bquote(&_msg_))+24)); 
 %put %str(ERR)OR: &_lin_; 
 %put %str(ERR)OR: &_msg_ Macro stoped processing.; 
 %put %str(ERR)OR: &_lin_; 
 %return; 

%end; 

 
2. READ PDF FILE INTO SAS 

At the first significant step, the macro reads the whole PDF file including the texts and the description of the PDF file 
format into a SAS dataset.  Then it extracts each text field and useful features as a string. The code is as below. 
 
%local _t_1 _t_2 _t_3 _rc_; 
%let _t_1 = \d+\.?\d*; 
%let _t_2 = ( (&_t_1) &_t_1 &_t_1 &_t_1 re B\*( \[\]0 d){0,2})?( &_t_1 &_t_1 
&_t_1 rg &_t_1 &_t_1 &_t_1 RG( &_t_1 w( \[\]0 d)? \d J \d j( \[\]0 d)?)?)?; 
%let _t_3 = &_t_2&_t_2&_t_2 BT\s?\/(TT|F)\d &_t_1 Tf (&_t_1) (&_t_1) Td 
\((.*)\)Tj ET; 
%let _rc_ = 4; 
options mprint noquotelenmax varlenchk=nowarn ls=140; 
 
%* Get values for variable String and Page *; 
data _tmp_1(keep=string page _p:); 
 length _S_0 $32767 string $1000; 
 array _S_[&_rc_] $32000; 
 retain _O_ 0 re page; 
 if _N_=1 then re = prxparse("/&_t_3/i"); 
    infile "&datafile..pdf" truncover lrecl=%eval(32000*&_rc_) end=eof; 
 input (_S_1-_S_&_rc_) ($char32000.); 
 _i_  = 1; 
 _p0_ = 768; 
 substr(_S_0,_p0_) = _S_[_i_]; 
 _p1_ = 1; 
 do until(_i_=&_rc_ & _p1_=0); 
  if _i_<&_rc_ & (_p0_>32000 | _p1_=0) then do; 
   _i_ + 1; 
   _p0_ = ifn(_p0_>32000,_p0_-32000,1); 
   substr(_S_0,_p0_) = substr(_S_0,_p0_+32000)||_S_[_i_]; 
  end; 
  _p1_ = find(_S_0,')Tj ET','i',_p0_); 
  if _p1_=0 then continue; 
  call prxnext(re,_p0_,_p1_+5,_S_0,_p1_,_L1_); 
  if _p1_=0 then continue; 
  if _N_>_O_ then page + 1; 
  string = substr(_S_0,_p1_,_L1_); 
  output; 
  _O_ = _N_; 
 end; 
 if eof then call symputx('_O_',_O_,'L'); 
run; 
%if &_O_=0 %then %do; 
%let _msg_ = Input PDF file (Empty/Compressed/Cryptographic?) 
"&datafile..pdf" is not recognized by macro.; 
%goto ER; 

%end; 

In the code above, the INFILE and INPUT statements read the PDF file into a SAS data set. The record length is very 
long because one record is obtained for a whole page of the texts and the related description of the PDF file format 
so we use 4 strings(_S_[1]  to _S_[4]) with length 32000 (total length 128000), then concatenate them and get useful 
texts and features. Here, the Perl Regular Expression function extracts the useful information into variable ‘string’ in 
output data set ‘_tmp_1’. Perl regular expression is a special text string for describing a search pattern. prxparse() 
and prxnext()  are Perl Regular Expression function in SAS. Another variable ‘page’ is for the page number. 

With the help of Perl Regular Expression as &_t_3 (in the code above), we extract the information into variable 
‘string’. For input PDF file ‘afmsg_Analysis.pdf’, the SAS dataset ‘_tmp_1’ is as follows: 

 
Display 2: In SAS data set ‘_tmp_1’, each record includes a text and formating information 
for each cell. 
 

3. EXTRACT TEXT AND ATTRIBUTES 

For the next step, we need to separate the text and each feature from the string. The following code can fulfill this 
goal. 

%* Get Loc1-Loc5 for location/format info of the page and Text *; 
data _tmp_2(drop=string i _Ps_); 
 array _R_[6] _temporary_; 
 if _N_=1 then do i=1 to 6; 
  _R_[i] = 
prxparse(cat("s/&_t_3/$",choosen(i,2,9,16,23,24,25),"/")); 
 end; 
 set _tmp_1; 
 by page; 
 array Loc[5]; 
 length Text $400; 
 do i=1 to 5; Loc[i] = input(prxchange(_R_[i],-1,string),best.); end; 
 if n(Loc1,Loc2,Loc3)>0 then if min(Loc1,Loc2,Loc3)<Loc4 then Loc0 = 
round(min(Loc1,Loc2,Loc3),2.5); 
 Text = prxchange(_R_[6],-1,string); 
 call prxposn(_R_[6],25,_Ps_,len); 

run; 

Similar to extract useful info into variable ‘string’, the Perl Regular Expression 
function is still used for separating the text and each attributes.  

Display 3: SAS data set ‘_tmp_2’ included the separated text and format information for 
each cell. 
In the SAS dataset _tmp_2, each cell’s text and features are represented in one observation. Each variable has the 
meaning as follows: 

(1) Loc1 to Loc3: One of them will be the x-coordinate for the cell's frame on the left side of the border (if 
having). 

(2) Loc4: The x-coordinate for the cell's text on the bottom- left of the border 
(3) Loc5: The y-coordinate for the cell's text 
(4) Variable Text: the text in each cell 
(5) Variable Len: the length (include the leading blank) of text in each cell 
(6) Variable Page: the page number where each cell is located in 

The unit of the length is point (1 inch equal to 72 point).  

4. OBTAIN ROW AND COLUMN NUMBER 

In SAS dataset ‘_temp_2’, the page number is the value of variable ‘Page’. But, we also need to get the column and 
row number for each cell in the page. By integrating the cell frame information, the row and column number can be 
obtained. The following is the code. 

%* Using proc sql for the row/column information on coordinate: Loc5, Loc0 *; 
proc sql noprint; 
 create table _rl_0 as select * from 
   (select page, Loc5, Loc0, n(1) as _xn_ from _tmp_2 group by page, Loc5) 
  group by page having _xn_=max(_xn_) order by page, Loc5 desc; 
 
 select _xn_ into :_xn_ trimmed from _rl_0(obs=1); 
quit; 
 
%* Get the x,y coordinator for the frame *; 
%local j; 
%do j=1 %to 2; 
proc sql; 
 create table _rl_&j as select distinct page, Loc%scan(5 0,&j), _xn_ 
   from _rl_0 %if &j=1 %then order by page, Loc5 desc; %else where 
Loc0>.;; 

quit; 
data _rl_%eval(&j+2); 
 set _rl_&j; 
 by page; 
  %if &j=1 %then %do; 
 length _ys_ $400; 
 retain _ys_; 
 if first.page then _ys_ = ''; 
 _ys_ = catx(' ',_ys_,Loc5); 
  %end; 
  %else %do; 
 array _x_[&_xn_]; 
 retain _x_1-_x_&_xn_; 
 if first.page then i = 0; 
 i + 1; 
 _x_[i] = Loc0; 
  %end; 
 if last.page; 
run; 
%end; 
 
 

 
Display 4: SAS data set ‘_rl_3’. Variable ‘_ys_’ is the list of row starting location for the y-coordinate 

Display 5: SAS data set ‘_rl_4’. Variable _x_1 - _x_5 listed the 5 column’s start location of x-coordinate. 

 

In the following code, data set _tmp_2 is merged with _rl_3 and _r1_4 to get the row and column number. 
%* According to the size frame, get row and column number *; 
data _tmp_3(keep=page Text len part row col); 
 merge _tmp_2 _rl_3(keep=page _ys_) _rl_4(keep=page _x_1-_x_&_xn_); 
 by page; 
 array _x_[&_xn_]; 
 retain part row col_ n_r L_y; 
 if first.page then do; 
  call missing(col_,row); 
  n_r = 1; 
  L_y = input(scan(_ys_,n_r,' '),best.); 
 end; 
 do i=1 to &_xn_; 
  if i<&_xn_ & Loc4>_x_[i+1] then continue; 
  col = i; 
  leave; 
 end; 
 if row=. then do; 
  if Loc5^=L_y then part = 1+(Loc5>250); 
  else do; row = 1;  part = 3; end; 
 end; 
 else if Loc5<L_y then do; 
  n_r + 1; 
  L_y = input(scan(_ys_,n_r,' '),best.); 
  if Loc5=L_y | Loc5>L_y & col=1<col_ then row + 1; 
 end; 
 else if Loc5>=L_y & col=1<col_ then row + 1; 
 col_ = col; 
run;  

 



 

 

8 

5. RE-CONSTRUCT TEXT FROM DATA SET 

After the number of rows and columns are obtained, a special situation needs to be considered. That is: the text is 
wrapped into a cell as multiple lines. Let’s see PDF file in Display 6. When the column TEXT is ‘Without the FSP 
product licensed, the Data Set Data Model(DATA_M) cannot be used with the Form Editor in build mode.’, the 
column TEXT is wrapped and the ‘TA_M) cannot be used with the Form Editor in build mode.’ is located in the 
second line of the cell which is shown on Display 6. The Re-construction is necessary for the dataset _tmp_3.  
 

 
Display 6: Sample  PDF output with the wrapped text in the cell 

The following is what happens in the dataset _tmp_3 in Display 7. Observe that both obs=1011 and 1012 belong to 
row 9 and column 5.   

 
Display 7: SAS data set ‘_tmp_3’. Column number and row number are obtained for text in each cell. 

So, the concatenation is necessary. The following code is for the concatenation. 

%* Combine the texts which is in same cell *; 
data _tmp_4; 
 set _tmp_3(where=(row>.)); 
 by page row col; 
 length Text_ $400; 
 retain Text_; 
 if first.col=1 & last.col=1 then; 
 else do; 
  if first.col=1 then do; 
   Text_ = Text; 
   _len_ = len; 
  end; 
  else do; 
   Text_ = substrn(Text_,1,_len_)||Text; 
   _len_ + len; 
  end; 
  if last.col=0 then delete; 
  else Text = Text_; 
 end; 
 Text = tranwrd(tranwrd(Text,'\050','('),'\(','('); 
 Text = tranwrd(tranwrd(Text,'\051',')'),'\)',')'); 
 Text = prxchange('s/\\177|\\000/�/',-1,Text); 
 Text = prxchange('s/\\134|\\\\/\\/',-1,Text); 
run; 
 
 
While we concatenate the wrapped text from dataset _tmp_3, it’s almost verbatim text for each cell except for 
following 4 characters: 

(1) Character ‘(’ --- open parenthesis, ASCII code 40, as 50 in octonary number system 
(2) Character ‘)’ --- closing parenthesis, ASCII code 41, as 51 in octonary number system 
(3) Character ‘�’ --- a character in ASCII code 127, as 177 in octonary number system  
(4) Character ‘\’ --- backslash, ASCII code 92, as 134 in octonary number system 

 
In PDF file, above 4 characters is presented as following way: 

(1) Character ‘(’ is presented as: (a) ‘\050’ by some style, or (b) ‘\(’ by some other style 
(2) Character ‘)’ is presented as: (a) ‘\051’ by some style, or (b) ‘\)’ by some other style 
(3) Character ‘�’ is presented as: (a) ‘\177’ by some style, or (b) ‘\000’ by some other style 
(4) Character ‘\’ is presented as: (a) ‘\134’ by some style, or (b) ‘\\’ by some other style 

 
In the code for generating dataset _tmp_4, the last four lines show how to convert the 4 kind of characters back to the 
original symbol. 

Display 8: SAS data set ‘_tmp_4’. Wrapped text are combined and the special characters ‘(’, ‘)’ are converted  

 
%* Keep the texts in the same row as the different variables *; 
proc transpose data=_tmp_4 out=_tmp_5 prefix=_; 
 by page row; 
 id col; 
 var Text; 
run; 
 
Look at the Display 7,8 in dataset _tmp_3 and _tmp_4, multiple consecutive observations have same row number but 
different column number, a transpose is necessary for the re-construction (to generate a dataset _tmp_5). 
 
 

 
Display 9: SAS data set ‘_tmp_5’. The data set is transposed by column numbers. 

%* Convert some the character variable as numeric *; 
data _tmp_6; 
 retain re; 
 if _N_=1 then re = prxparse('/^\s*-?\d{0,20}\.?\d{0,20}\s*$/'); 
 set _tmp_5 end=eof; 
 by page row; 
 array _A_[*] $400 _1-_&_xn_; 
 array _V_[*] _V_1-_V_&_xn_; 
 array _x_[&_xn_] _temporary_(&_xn_*0); 
 array _L_[&_xn_] _temporary_(&_xn_*0); 
 array _H_[&_xn_] $32 _temporary_; 
 length _Var_ $32 _Keep_ _Rename_ _Length_ $1000; 
 if first.page then do; 
  if page=1 then do i= 1 to &_xn_; 
   _H_[i] = translate(cats(_A_[i]),'______','(-/ %)'); 
  end; 
  delete; 
 end; 
 else do i=1 to &_xn_; 
  _L_[i] = max(_L_[i],length(_A_[i])); 
  if _x_[i]>0 then continue; 
  if prxmatch(re,_A_[i])=0 then _x_[i] + 1; 
  else _V_[i] = input(_A_[i],best12.); 
 end; 
 if eof then do; 
  do i=1 to &_xn_; 
   _Fc_ = 50**(_L_[i]>32); 
   _Var_ = catt(ifc(_x_[i]=0,'_V_','_'),i); 
   _Keep_ = catx(' ',_Keep_,_Var_); 
   _Rename_= catx(' ',_Rename_,_Var_,'=',_H_[i]); 
   _length_= catx(' 
',_length_,_H_[i],ifc(_x_[i]=0,'8',cat('$',_Fc_*2**ceil(log2(_L_[i]/_Fc_)))))
; 
  end; 
  call symputx('_Keep_', _Keep_,  'L'); 
  call symputx('_Rename_',_Rename_,'L'); 
  call symputx('_Length_',_Length_,'L'); 
 end; 
run; 
 
In the code above for data set _tmp_6, the first thing is to separate the first row in each pages as the header for the 
name of each column/variable. Then, please notice that the PDF present a numeric value as the character in numeric 
form. In _tmp_5, all text-related variables are in character. In code for dataset _tmp_6, we are looking for the variable 
with all value as numeric form, and convert them into numerical variables.  
 
%* Set up the data set label, the variable name and suitable length *; 
data %str(&out)(label="%scan(&datafile,-1,\/)"); 
 length &_Length_; 
 set _tmp_6(keep=&_Keep_ rename=(&_Rename_)); 
run; 
And also, the variable Text has initial length 400, from output SAS data set, each individual character variables get 
his suitable length by setting a length statement. 
 
The combined code sections above is the entire program of macro %PDF2SAS. 

CONCLUSION 
The SAS Output Delivery System (ODS) is a powerful system to help manipulate and customize outputs.  But, this 
technique is only one-way (or called single trip). By means of the Perl Regular Expression, a whole PDF file can be 
successfully separated into many pieces where there are important information to re-construct them back into a SAS 
data set. The macro %PDF2SAS can make the PDF file easily and quickly to be converted to SAS datasets and to be 
validated. 

Appendix 1: 
 
%global _Name_ _o_; 
%let _Name_ = sashelp.afmsg; 
 
ods _all_ close; 
ods output Template.Stats=Stats(where=(Type='Style')); 
proc template; 
 list styles; 
run; 
ods output close; 
ods listing; 
 
proc sql noprint; 
 select distinct cat(scan("&_Name_",-1,'.'),'_',scan(Path,-1,'.')) into 
:_f_1-:_f_99 
   from Stats where index(Path,'NoFontDefault')=0; 
 %let _o_ = &sqlobs; 
quit; 

http://www.adobe.com/content/dam/Adobe/en/devnet/acrobat/pdfs/pdf_reference_1-7.pdf
http://www.adobe.com/content/dam/Adobe/en/devnet/acrobat/pdfs/pdf_reference_1-7.pdf
mailto:willywu2001@hotmail.com

%macro getpdf; 
title1 j= 
c 'SAS Help Message Listing'; 
footnote1 j=l 'This is a sample.'; 
ods escapechar=' '; 
 
ods listing close; 
%do i=1 %to &_o_; 
ods pdf file="&path\PDF\&&_f_&i...pdf" style=%scan(&&_f_&i,-1,_) compress=0; 
proc report nowd data=&_Name_ style(column hdr)={asis=on} split=' '; 
 column MSGID MNEMONIC LINENO LEVEL TEXT PBUTTONS; 
 define MSGID /display style=[cellwidth=0.7in]; 
 define MNEMONIC /display style=[cellwidth=2.0in]; 
 define LINENO /display style=[cellwidth=0.7in]; 
 define LEVEL /display style=[cellwidth=0.7in]; 
 define TEXT  /display style=[cellwidth=3.8in]; 
 define PBUTTONS /display style=[cellwidth=1.6in]; 
run; 
ods pdf close; 
%end; 
ods listing; 
%mend getpdf; 
 
%getpdf; 
title; 
footnote; 

